# Maintainer: Andrea Scarpino <andrea@archlinux.org>

pkgname=fastjar
pkgver=0.98
pkgrel=6
pkgdesc="Sun java jar compatible archiver"
arch=('x86_64')
url='https://savannah.nongnu.org/projects/fastjar'
license=('GPL')
depends=('zlib' 'texinfo')
source=(https://download.savannah.nongnu.org/releases/${pkgname}/${pkgname}-${pkgver}.tar.gz)
# The key cannot be found on any keyserver. Keyring can only be downloaded here
# https://savannah.nongnu.org/project/memberlist-gpgkeys.php?group=fastjar
#validpgpkeys=('EEFF99F6ADE26B135C85141D663843B80213D86E') # Dalibor Topic
sha256sums=('f156abc5de8658f22ee8f08d7a72c88f9409ebd8c7933e9466b0842afeb2f145')

build() {
  cd ${pkgname}-${pkgver}
  ./configure --prefix=/usr
  make
}

package() {
  cd ${pkgname}-${pkgver}
  make DESTDIR="${pkgdir}" install
}
